DECUS
PROGRAM LIBRARY

DECUS NO.	8-655
TITLE	PATCHES TO CINET-BASIC (DECUS NO. 8-159)
AUTHOR	Geoffrey Chase
COMPANY	Portsmouth Abbey School Portsmouth, Rhode Island
DATE	August 30, 1973
SOURCE LANGUAGE	PAL III

/PATCHES TO CINET-BASIC 8/73

DECUS Program Library Write-up DECUS NO. 8-655

A. THE FIRST PATCH REPLACES THE INTERNAL 'ALIGN' AND 'FIX' ROUTINES WITH NEWER VERSIONS LIFTED FROM THE CURRENT FOCAL-8. THE " INT(X) " FUNCTION IS NOW FREED FROM ANY RESTRICTION ON THE SIZE OF ITS ARGUMENT, X. THUS " LET Y=INT(X) " WHERE X = 10999.7 WILL WORK PERFECTLY WELL AND WILL NOT TRIGGER AN ERROR DIAGNOSTIC.

THIS PATCH OVERLAYS EXISTING CORE USED BY THE FLOATING-POINT ROUTINES AND IS THUS INVISIBLE TO THE USER. NO SACRIFICE IS MADE BY LOADING IT.

B. THE SECOND PATCH IS A COMPROMISE. IT REENABLES THE " SGN(X) " FUNCTION BUT HAS TWO DRAWBACKS:

1. " SGN(X) " ALWAYS RETURNS WITH -1 (IF X<0) OR +1 (IF X>=0) THUS SGN(0) = 1 [NOT 0]. THIS DIFFERS FROM STANDARD BASIC. IN MOST CASES THE DIFFERENCE DOES NOT MATTER MUCH, AND IT DOES CORRESPOND TO THE INTERNAL PDP-8 LOGIC, WHICH CONSIDERS 0 TO BE "POSITIVE".

2. THE PATCH USES LOCATIONS 7600-7610. IN A PAPER-TAPE SYSTEM THESE ARE FREE, AND THUS COST THE USER NOTHING. I HAVE ASSUMED MOST CINET USERS HAVE PAPER-TAPE SYSTEMS, SINCE DISK ETC. OPERATING SYSTEMS WILL RUN MUCH MORE POWERFUL VERSIONS OF BASIC THAN CINET -- E.G., THE DIALECTS OF EDU-30 FROM DEC OR OMSI.

IF YOU DO RUN CINET UNDER A DISK OR TAPE OPERATING SYSTEM WITH THIS PATCH INCLUDED, IT WILL BE NECESSARY TO RE-BOOTSTRAP THE SYSTEM WHEN YOU ARE DONE.

REMARKS:

1. THESE PATCHES ARE AT BEST PALLIATIVES. SOMEONE REALLY OUGHT TO GO THROUGH THE WHOLE FLOATING-POINT PROCESSOR AND SORTING ROUTINES TO UPDATE THEM. CINET-BASIC IS ESSENTIALLY A COPY OF AN EARLY VERSION OF FOCAL, WHICH HAS BEEN MUCH IMPROVED SINCE.

2. SOMETHING IS NOT QUITE RIGHT IN THE INTERPRETER/EVLUATOR ALGORITHMS. THIS USER, AT LEAST, HAS FOUND FOR...NEXT A BIT UNPREDICTABLE.

3. SOME OF THE ODDITIES -- "IF X<0" WON'T WORK, BUT "IF X<0." AND "IF X<1" (NO PERIOD) WILL WORK -- COULD PROBABLY BE GOT OUT.

STILL, IT IS A WORTHWHILE LITTLE LANGUAGE FOR A 4-K PAPER TAPE SYSTEM AND, LIKE ALL DECUS OFFERINGS, REMARKABLY INEXPENSIVE.

APPENDED LISTINGS SUGGEST SOME USES FOR THESE FUNCTIONS. SGN(X) IS POPULAR IN BRANCHING COMMANDS--FOR EXAMPLE, A BRUTE-FORCE SEARCH FOR ROOTS TO AN EQUATION MIGHT LOOK FOR A CHANGE OF SIGN IN Y TO BRACKET AN INTERVAL CONTAINING (PROBABLY) A ROOT VALUE OF X (ONE WHICH CAUSES Y TO BE 0).

```
                /PATCHES TO CINET-BASIC
                /FLOATING-PT. PROCESSOR: 'FIX' AND 'ALIGN'

                FLAC=EXP=44
                HORD=   45
                LORD=   46
                OVER2=  47

                EX1=    40      /USED FOR ARGUMENT, E.G. ADDENDUM
                AC1H=   41
                AC1L=   42
                OVER1=  43

                ACMINS= 6600    /NEGATES AC; ENTRY PT. USED ALSO
                                /AS TEMP. STORAGE.

                *6620

6620  0000      ALIGN,  0       /SUBR. TO ALIGN BINARY POINTS
6621  1045              TAD HORD
6622  7450              SNA
6623  1046              TAD LORD
6624  7650              SNA CLA
6625  5306              JMP NOX1        /MANTISSA OF FLAC=0
6626  1041              TAD AC1H
6627  7450              SNA
6630  1042              TAD AC1L
6631  7450              SNA
6632  1043              TAD OVER1
6633  7650              SNA CLA
6634  5620              JMP I ALIGN     /ADDENDUM IS 0, EXIT
6635  1040              TAD EX1
6636  7041              CIA
6637  1044              TAD EXP
6640  7450              SNA
6641  5270              JMP ADONE       /EXPONENTS ARE EQUAL
6642  3200              DCA ACMINS
6643  1200              TAD ACMINS
6644  7500              SMA
6645  7041              CIA
6646  3317              DCA AMOUNT      /UNEQUAL, SAVE -DIFFERENCE
6647  1317              TAD AMOUNT
6650  1354              TAD TEST2
6651  7710              SPA CLA
6652  5272              JMP NOX         /EXPONENTS CANNOT BE ALIGNED
6653  1200              TAD ACMINS      /CAN BE: SHIFT THE SMALLER
6654  7700              SMA CLA
6655  5262              JMP ASHFT
6656  4721              JMS I PDIV2
6657  2317              ISZ AMOUNT
6660  5256              JMP .-2
6661  5270              JMP ADONE
```

```
6662  7040  ASHFT,  CMA
6663  1040          TAD EX1
6664  3040          DCA EX1
6665  4720          JMS I PDIV1
6666  2317          ISZ AMOUNT
6667  5265          JMP .-2
6670  2220  ADONE,  ISZ ALIGN        /RETURN TO CALL+2
6671  5620          JMP I ALIGN
6672  1040  NOX,    TAD EX1          /MISSION IMPOSSIBLE
6673  7700          SMA CLA
6674  5301          JMP NOX2
6675  1044          TAD EXP
6676  7700          SMA CLA
6677  5620          JMP I ALIGN      /TO CALL+1
6700  5303          JMP .+3
6701  1044  NOX2,   TAD EXP
6702  7700          SMA CLA
6703  1200          TAD ACMINS       /TEMP. STORAGE OF DIFFERENCE;
                                     /BOTH EXP. + OR BOTH EXP. -
6704  7740          SMA SZA CLA
6705  5620          JMP I ALIGN      /TO CALL+1
6706  1040  NOX1,   TAD EX1
6707  3044          DCA EXP
6710  1041          TAD AC1H
6711  3045          DCA HORD
6712  1042          TAD AC1L
6713  3046          DCA LORD
6714  1043          TAD OVER1
6715  3047          DCA OVER2
6716  5620          JMP I ALIGN      /AGAIN, TO CALL+1
6717  0000  AMOUNT, 0
6720  7222  PDIV1,  7222     /CALLED DIV2 IN CINET, DIV1 IN FOCAL
6721  7007  PDIV2,  7007     /CALLED DIV1 IN CINET, ETC.

              /SUBROUTINES TO TAKE ABSOLUTE VALUE, SAVE SIGN,
              /AND TO RESTORE ORIGINAL SIGN

              MINSKI= 53       /FOR REFERENCE, THIS IS PTR. TO "ACMINS"

6722  0000  ABSOLV, 0
6723  1045          TAD HORD
6724  3336          DCA SIGNF
6725  1045          TAD HORD
6726  7710          SPA CLA
6727  4200          JMS ACMINS       /IT WAS NEG., NEGATE IT
6730  5722          JMP I ABSOLV
6731  0000  RESOLV, 0
6732  1336          TAD SIGNF
6733  7710          SPA CLA
6734  4200          JMS ACMINS       /RESTORE ORIGINAL MINUS SIGN
6735  5731          JMP I RESOLV

6736  0000  SIGNF,  0        /HOLDS ORIGINAL SIGN OF FLAC MANTISSA
```

```
6737  0000  ZTEM1,  0          /AVAILABLE FOR RENT
6740  0000  SWITCH, 0          /USED W. "FIX" WHEN CALLED BY  INT(X) .

6741  0000  FIX,    0          /LEAVES 12-BIT INTEGER IN HORD, EXCEPT
                                /WHEN CALLED FROM "XINT" [INT(X)], WHEN
                                /INSTEAD IT LEAVES FLAC AS A FLTNG. INT.
6742  3340          DCA SWITCH       /"XINT" WILL LEAVE 7777 IN SWITCH
6743  4322          JMS ABSOLV
6744  1044          TAD EXP          /TEST FOR FRACTION
6745  7750          SPA SNA CLA
6746  5373          JMP FIXM         /DOUBLE-CHECK FOR -1
6747  7001          IAC
6750  3043          DCA OVER1
6751  1372          TAD P27
6752  3040          DCA EX1
6753  4220          JMS ALIGN
6754  0027  TEST2,  27
6755  2047          ISZ OVER2
6756  5362          JMP OVERKL
6757  2046          ISZ LORD
6760  7410          SKP
6761  2045          ISZ HORD
6762  3047  OVERKL, DCA OVER2        /CLEAR THE FRACTION
6763  4331          JMS RESOLV
6764  2340           ISZ SWITCH      /CALL FROM "XINT" ?
6765  7410           SKP
6766  5506           JMP I EFUN3I    /YES, EXIT DIRECTLY
6767  1046          TAD LORD
6770  3045          DCA HORD         /OLD F.P. PACKAGE EXPECTS THIS
6771  5741          JMP I FIX
6772  0027  P27,    27

6773  3044  FIXM,   DCA EXP          /ZERO OUT ALL OF FLAC
6774  3045          DCA HORD
6775  3046          DCA LORD
6776  5362          JMP OVERKL

              *1156

1156  7340  XINT,   CLA CLL CMA     /7777
1157  4452          JMS I INTEGE     /NEVER RETURNS!

              /REST OF OLD "XINT" IS USED AS EXIT BY RND(X) ["XRAN"]

              EFUN3I= 106      /POINTER TO RETURN FROM A FUNCTION
              INTEGER=52       /POINTS TO "FIX"
```

/SYMBOL TABLE:

```
ABSOLV  6722
ACMINS  6600
AC1H    0041
AC1L    0042
ADONE   6670
ALIGN   6620
AMOUNT  6717
ASHFT   6662
EFUN3I  0106
EXP     0044
EX1     0040
FIX     6741
FIXM    6773
FLAC    0044
HORD    0045
INTEGE  0052
LORD    0046
MINSKI  0053
NOX     6672
NOX1    6706
NOX2    6701
OVERKL  6762
OVER1   0043
OVER2   0047
PDIV1   6720
PDIV2   6721
P27     6772
RESOLV  6731
SIGNF   6736
SWITCH  6740
TEST2   6754
XINT    1156
ZTEM1   6737
```

```
                /SGN(X) PATCH
                /RESTORES AND REPAIRS FUNCTION IN CINET-BASIC
                /OVERLAYS LOCATIONS 7600-7610.  USERS OF OPERATING
                /SYSTEMS MAY FIND THIS UNDESIRABLE [THE PAPER TAPE
                /BINARY LOADER IS NOT AFFECTED].

                / 8/29/73

                /DEFINITIONS:

                XSGN=   2027    /ATTEMPTS TO OVERLAY THIS FAILED
                XXSGN=  7600
                FLAC=   44
                HORD=   45      /HI-ORDER MANTISSA WORD OF FLAC
                PUSHF=  4512    /PUSH ONTO FLOATING STACK
                FLTONE= 2366
                POPF=   4513    /INVERSE OF PUSHF
                EFUN3I= 106     /POINTER TO FUNCTION EXIT
                ABSOLV= 6722    /THE FLTG.-PT./INT(X) PATCH MUST
                RESOLV= 6731    /   BE LOADED; THESE ARE PART OF IT.

                *403
0403    7600            XXSGN   /ENABLE SGN(X) IN FNTABF

                *XXSGN
7600    4607            JMS I PABSLV    /GETS ABS. VALUE & SAVES SIGN
7601    4512            PUSHF
7602    2366    ARG     FLTONE
7603    4513            POPF
7604    0044    ARG     FLAC
7605    4610            JMS I PRESLV    /RESTORES ORIGINAL SIGN
7606    5506            JMP I EFUN3I

7607    6722    PABSLV, ABSOLV
7610    6731    PRESLV, RESOLV


                /THIS "SGN(X)" HAS THE PECULIARITY THAT SGN(0)=1 (POS.)
                /RATHER THAN 0, AS IN MOST BASICS.  SPACE PREVENTS A FIX,
                /BUT NORMALLY THIS PDP-8 STYLE "POSITIVE" DOES LITTLE HARM.

                /SYMBOL TABLE:

ABSOLV  6722
EFUN3I  0106
FLAC    0044
FLTONE  2366
HORD    0045
PABSLV  7607
POPF    4513
PRESLV  7610
PUSHF   4512
RESOLV  6731
XSGN    2027
XXSGN   7600
```

```
   CINET-BASIC
0010 PRINT
0020 PRINT "PRIMES"
0030 PRINT "------"
0040 REM:          DEMO OF EXTENDED "INT" FUNCTION IN ACTION
0100 LET N=49999
0110 LET N=N+2,D=1,S=SQR(N)
0115 IF N>50100 THEN 200
0120 LET D=D+2
0130 REM:          ODD NOS. & DIVISORS; SAVES TIME
0140 IF D>S THEN 180
0150 LET Q=N/D
0160 IF Q=INT(Q) THEN 110
0170 GOTO 120
0180 PRINT N
0190 GOTO 110
0200 END

RUN

PRIMES
------
 50021.
 50023.
 50033.
 50047.
 50051.
 50053.
 50069.
 50077.
 50087.
 50093.

READY

REM. #1: ABOUT 1 MINUTE
REM. #2: SAD EXPERIENCE WITH "FOR" LOOPS LED TO THEIR OMISSION
```

```
LIS
   CINET-BASIC
0010 REM:        PROGRAM TO ROUND NUMBERS.  USES NEW "INT" & "SGN"
0020 PRINT
0025 PRINT "VALUE          ROUNDED"
0030 PRI
0035 LET X=-8044.4
0040 GOSUB 1000
0050 LET X=ABS(X)
0060 GOSUB 1000
0070 LET X=-8044.5
0080 GOS 1000
0090 LET X=-X
0100 GOS 1000
0110 FOR X=-1/8 TO 1/8 STEP 1/32
0120 GOS 1000
0130 NEXT X
0140 STOP
0150 REM:--------
1000 PRINT X, INT(X+SGN(X)/2
1010 RETURN
```

```
RUN

VALUE           ROUNDED

-8044.4         -8044.
 8044.4          8044.
-8044.5         -8045.
 8044.5          8045.
-Ø.125           Ø.
-Ø.ØØ375         Ø.
-Ø.Ø625          Ø.
-Ø.Ø3125         Ø.
 Ø.              Ø.
 Ø.Ø3125         Ø.
 Ø.Ø625          Ø.
 Ø.ØØ375         Ø.
 Ø.125           Ø.

READY

PRINT SGN(Ø)
 1.

REMARK: SGN(Ø)=1 [POSITIVE], WHICH IS A DIFFERENCE FROM THE
REM     MORE COMMON BASICS.  IN MOST USES, SUCH AS THIS ONE,
REM     IT MAKES LITTLE DIFFERENCE.


1ØØØ PRINT X,INT(2*X+SGN(X)/2)/2

REM     THIS WILL ROUND NUMBERS TO THE NEAREST HALF.  TO ROUND TO
REM     THE NEAREST 1Ø'TH, CHANGE FIRST & LAST 2'S TO 1Ø'S.

RUN

VALUE           ROUNDED

-8044.4         -8044.5
 8044.4          8044.5
-8044.5         -8044.5
 8044.5          8044.5
-Ø.125           Ø.
-Ø.ØØ375         Ø.
-Ø.Ø625          Ø.
-Ø.Ø3125         Ø.
 Ø.              Ø.
 Ø.Ø3125         Ø.
 Ø.Ø625          Ø.
 Ø.ØØ375         Ø.
 Ø.125           Ø.

READY
```

```
LIS
   CINET-BASIC
0050 REM:  DEMO OF BRUTE FORCE SEARCH, USING "INT" & "SGN"
0060 REM:  OBVIOUSLY SUCH A SIMPLE QUADRATIC IS BETTER SOLVED
0070 REM:  DIRECTLY, BUT THE TECHNIQUE CAN BE USED FOR HIGHER
0080 REM:  ORDER EQNS.
0090 REM
0100 LET L=-10,H=10
0110 LET I=(H-L)/20
0115 PRINT "L=",L,"H=",H
0120 LET X=L-I
0130 LET X=X+I
0140 LET Y=X↑2+2*X+.5
0142 LET Z=INT(1E5*Y+SGN(Y)/2)/1E5
0145 IF Z=0. THEN 290
0150 IF X=L THEN 200
0160 IF P=SGN(Y) THEN 130
0170 LET L=X-I,H=X
0180 GOTO 110
0200 LET P=SGN(Y)
0210 GOTO 130
0290 PRINT
0300 PRINT "X=",X,"Y=",Y
0310 END


RUN
L=-10.          H= 10.
L=-2.           H=-1.
L=-1.75         H=-1.7
L=-1.7075       H=-1.705
L=-1.70713      H=-1.707

X=-1.70711      Y=-0.19073  E-05

READY


REMARK " 160 IF SGN(Y)=P THEN 130 " DIDN'T WORK; SIMILARLY
REM    WITH LINE 145--HENCE THE ADDITION OF LINE 142.
```